/*
 *  Copyright (c) 2010 Brian Stanek <bstanek@gmail.com>
 *
 *  Permission is hereby granted, free of charge, to any person
 *  obtaining a copy of this software and associated documentation
 *  files (the "Software"), to deal in the Software without
 *  restriction, including without limitation the rights to use,
 *  copy, modify, merge, publish, distribute, sublicense, and/or sell
 *  copies of the Software, and to permit persons to whom the
 *  Software is furnished to do so, subject to the following
 *  conditions:
 *
 *  The above copyright notice and this permission notice shall be
 *  included in all copies or substantial portions of the Software.
 *
 *  THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,
 *  EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES
 *  OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND
 *  NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT
 *  HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY,
 *  WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING
 *  FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR
 *  OTHER DEALINGS IN THE SOFTWARE.
 */

package guice.utils
{

import flash.errors.IllegalOperationError;
import flash.utils.Proxy;
import flash.utils.flash_proxy;

/**
 *  An unordered, unduplicated list.
 */
public class Set extends Proxy
{

    //--------------------------------------------------------------------------
    //
    //  Constructor
    //
    //--------------------------------------------------------------------------

    public function Set()
    {
        super();
    }

    //--------------------------------------------------------------------------
    //
    //  Variables
    //
    //--------------------------------------------------------------------------

    /**
     *  @private
     */
    private var elements:Array = [];

    //--------------------------------------------------------------------------
    //
    //  Properties
    //
    //--------------------------------------------------------------------------

    //----------------------------------
    //  length
    //----------------------------------

    public function get length():int
    {
        return elements.length;
    }

    //--------------------------------------------------------------------------
    //
    //  Methods
    //
    //--------------------------------------------------------------------------

    public function add(element:Object):Boolean
    {
        if (!contains(element))
        {
            elements.push(element);
            return true;
        }
        return false;
    }

    public function contains(element:Object):Boolean
    {
        return elements.indexOf(element) != -1;
    }

    //--------------------------------------------------------------------------
    //
    //  Methods: Proxy
    //
    //--------------------------------------------------------------------------

    /**
     *  @private
     */
    override flash_proxy function nextNameIndex(index:int):int
    {
        return index < elements.length ? index + 1 : 0;
    }

    /**
     *  @private
     */
    override flash_proxy function nextName(index:int):String
    {
        throw new IllegalOperationError("Set does not support indexed iteration: use 'for each...in' instead");
    }

    /**
     *  @private
     */
    override flash_proxy function nextValue(index:int):*
    {
        return elements[index - 1];
    }

}

}
